Programmation Fonctionnelle Avancée 8 : Inférence de types, polymorphie et traits impératifs

Ralf Treinen

Université Paris Diderot
UFR Informatique
Institut de Recherche en Informatique Fondamentale

PARIS DIDEROT

treinen@irif.fr

7 novembre 2018

Rappel sur le typage en OCaml

Les deux traits essentiels du système de typage de OCaml sont :

- Système de types *polymorphe* : List.map manipule des listes de tout type. Les listes sont polymorphes, mais homogènes (dans chaque liste, tous les éléments ont le même type).
- *Inférence de types* : le système découvre tout seul le type le plus général, sans besoin de déclarer les types des identificateurs.

Exemples (inf1.ml)

```
List.map;;

List.map (fun x -> x + 1) [1;2;3];;

List.map (fun s -> s^"-") ["a";"b";"c"];;
```


Exemples (inf2.ml)

```
let k x y = x;;

let s x y z = (x y) (y z);;

let f x y z = x (y z) (y,z);;
```

Inférence de types

- Comment est-ce que OCaml fait pour trouver le type le plus général d'un identificateur ?
- Regardons d'abord le dernier exemple du transparent précédant.
- Il s'agit d'un cas simple : pas de récurrence.
- On introduit une variable pour le type de chaque identificateur nouveau (ici : pour les identificateurs f, x, y, z), et une variable pour chacune des expressions du côté droite :

$$\texttt{let } f\ x\ y\ z\ = \overbrace{x\ \underbrace{(y\ z)}_{e_2}\ \underbrace{(y,z)}_{e_3}}^{e_1}$$

- Variables : $t_f, t_x, t_y, t_z, t_1, t_2, t_3$

Système d'équations entre types

- Rappel : $\rightarrow$ associe à droite, c.-à-d. :

$$x \rightarrow y \rightarrow z \qquad \text{est à lire comme} \qquad x \rightarrow (y \rightarrow z)$$

- Pour **let** f x1 ... xn = c :
  - $t_f = t_{x1} \rightarrow \ldots \rightarrow t_{xn} \rightarrow t_c$
- Pour tous les sous-expressions e de c, c incluse :
  - si $e = (e1, e2)$: $t_e = t_{e1} \times t_{e2}$
  - si $e = e1\ e2\ \ldots en$: $t_{e1} = t_{e2} \rightarrow \ldots t_{en} \rightarrow t_e$

Systèmes d'équations entre types

- On a :

$$\texttt{let } f\ x\ y\ z\ = \overbrace{x\ \underbrace{(y\ z)}_{e_2}\ \underbrace{(y,z)}_{e_3}}^{e_1}$$

- On écrit des équations, on utilisant les règles d'associativité :

$$\begin{aligned} t_f &= t_x \rightarrow (t_y \rightarrow (t_z \rightarrow t_1)) \\ t_x &= t_2 \rightarrow (t_3 \rightarrow t_1) \\ t_y &= t_z \rightarrow t_2 \\ t_3 &= t_y \times t_z \end{aligned}$$

Système d'équations entre types

- Comment obtenir t_f à partir de ces équations ?
- On peut d'abord réécrire le système d'équations en une forme plus habituelle :

$$\begin{aligned} t_f &= f(t_x, f(t_y, f(t_z, t_1))) \\ t_x &= f(t_2, f(t_3, t_1)) \\ t_y &= f(t_z, t_2) \\ t_3 &= p(t_y, t_z) \end{aligned}$$

où $f(x, y)$ remplace $x \rightarrow y$, et $p(x, y)$ remplace $x \times y$.

Le sens des équations entre types

- Dans les équations il y a des variables, et des constructeurs de types f ($\rightarrow$) et p ($\times$), et éventuellement des constantes (`int`, `bool`).
- Propriétés des constantes et constructeurs :
  - Deux termes avec des constructeurs/constantes différentes à la tête ne peuvent jamais être égaux.
  - $p(x_1, x_2) = p(y_1, y_2)$ exactement si $x_1 = y_1$ et $x_2 = y_2$. Pareil pour f.
- Ce sont précisément les lois des symboles de fonctions non interprétées comme on les connaît de la Logique !

Resolution d'équations entre types

- L'algorithme pour résoudre des équations entre termes dans une structure de symboles de fonctions non interprétées est précisément l'algorithme d'*unification* de Herbrand (voir un cours de *Logique*) !
- L'unification nous donne soit l'information que le système d'équations n'a pas de solution, soit une solution la plus générale (mgu) : toute solution peut être obtenue comme instance du mgu.
- Nous cherchons le type le plus général de f qui est permis par la définition de f, cela correspond exactement au mgu du système des équations.

Rappel : L'algorithme d'unification (1)

- Donné :
  - une signature Σ (un ensemble de symboles de fonction avec leur arité). Dans le cas des équations de types on a

    $$\Sigma = \{p, f, \texttt{int}, \texttt{bool}, \ldots\}$$

    où p, f ont arité 2, `int` et `bool` ont arité 0.
  - Un ensemble V de variables.
- $T(\Sigma, V)$: l'ensemble des termes construits sur Σ et V
- $free(t)$: l'ensemble des variables qui paraissent dans le terme t

Rappel : L'algorithme d'unification (2)

- Règles de transformation sur un système d'équations.
- *Decomposition* :

  $$\frac{g(s_1, \ldots, s_n) = g(t_1, \ldots, t_n)}{s_1 = t_1, \ldots, s_n = t_n}$$

- *Clash* :

  $$\frac{g(s_1, \ldots, s_n) = h(t_1, \ldots, t_m)}{false}$$

  quand g différent de h

Rappel : L'algorithme d'unification (3)

- *Occur Check* :

$$\frac{x = t}{\mathit{false}}$$

quand $x \in \mathit{free}(t)$, et t est différent de x.

- *Variable Elimination* :

$$\frac{x = t \wedge \phi}{x = t \wedge \phi[x/t]}$$

quand $x \notin \mathit{free}(t)$, $x \in \mathit{free}(\phi)$.

Rappel : L'algorithme d'unification (4)

- *Variable Orientation*

$$\frac{t = x}{x = t}$$

quand t n'est pas une variable

- *Trivial Equation*

$$\frac{x = x}{\mathit{true}}$$

Rappel : L'algorithme d'unification (5)

- Cet algorithme termine toujours : soit il donne *false*, soit un système d'équations en forme normale (aucune transformation ne s'applique).
- Le résultat est *équivalent* au système d'origine.
- Quand l'algorithme se termine avec un résultat différent de *false* : on a un système d'équations de la forme

$$\begin{aligned} x_1 &= t_1 \\ &\vdots \\ x_n &= t_n \end{aligned}$$

où aucun des x_i parait dans les termes t_j.

- Variantes : séparation entre équations résolues et non résolution, calcul d'une solution sous forme triangulaire.

Exemple : résolution d'équations (1)

- Système de départ :

$$\begin{aligned} t_f &= f(t_x, f(t_y, f(t_z, t_1))) \\ t_x &= f(t_2, f(t_3, t_1)) \\ t_y &= f(t_z, t_2) \\ \underline{t_3} &= p(t_y, t_z) \end{aligned}$$

- Après élimination de t_3 :

$$\begin{aligned} t_f &= f(t_x, f(t_y, f(t_z, t_1))) \\ t_x &= f(t_2, f(p(t_y, t_z), t_1)) \\ t_y &= f(t_z, t_2) \\ t_3 &= p(t_y, t_z) \end{aligned}$$

Exemple : résolution d'équations (2)

- Après élimination de t_3 :

$$\begin{aligned} t_f &= f(t_x, f(t_y, f(t_z, t_1))) \\ t_x &= f(t_2, f(p(t_y, t_z), t_1)) \\ \underline{t_y} &= f(t_z, t_2) \\ t_3 &= p(t_y, t_z) \end{aligned}$$

- Après élimination de t_y :

$$\begin{aligned} t_f &= f(t_x, f(f(t_z, t_2), f(t_z, t_1))) \\ t_x &= f(t_2, f(p(f(t_z, t_2), t_z), t_1)) \\ t_y &= f(t_z, t_2) \\ t_3 &= p(f(t_z, t_2), t_z) \end{aligned}$$

Exemple : résolution d'équations (3)

- Après élimination de t_y :

$$\begin{aligned} t_f &= f(t_x, f(f(t_z, t_2), f(t_z, t_1))) \\ \underline{t_x} &= f(t_2, f(p(f(t_z, t_2), t_z), t_1)) \\ t_y &= f(t_z, t_2) \\ t_3 &= p(f(t_z, t_2), t_z) \end{aligned}$$

- Après élimination de t_x :

$$\begin{aligned} t_f &= f(f(t_2, f(p(f(t_z, t_2), t_z), t_1)), f(f(t_z, t_2), f(t_z, t_1))) \\ t_x &= f(t_2, f(p(f(t_z, t_2), t_z), t_1)) \\ t_y &= f(t_z, t_2) \\ t_3 &= p(f(t_z, t_2), t_z) \end{aligned}$$

Exemple : résolution d'équations (4)

- Le mgu associe à t_f le terme :

$$f(f(t_2, f(p(f(t_z, t_2), t_z), t_1)), f(f(t_z, t_2), f(t_z, t_1)))$$

- C-à-d, dans la notion habituelle de OCaml, le type de f est :

$$((t_2 \to (t_z \to t_2) \times t_z) \to t_1) \to (t_z \to t_2) \to t_z \to t_1$$

- où, après renommage des variables ($t_2 \mapsto a$, $t_z \mapsto b$, $t_1 \mapsto c$) :

$$(a \to (b \to a) \times b) \to c) \to (b \to a) \to b \to c$$

Exemple d'échec de l'inférence de type

- Regardons un deuxième exemple :

```
let f g = (g 42) && (g "coocoo")
```

- Système d'équations :

$$\begin{aligned} t_f &= f(t_g, \texttt{bool}) \\ t_g &= f(\texttt{int}, \texttt{bool}) \\ t_g &= f(\texttt{string}, \texttt{bool}) \end{aligned}$$

- On obtient avec les règles d'unification :

$$\begin{aligned} f(\texttt{int}, \texttt{bool}) &= f(\texttt{string}, \texttt{bool}) \\ \texttt{int} &= \texttt{string} \quad \lightning \end{aligned}$$

Pourquoi cet echec ?

- `let f g = (g 42) && (g "coocoo")`
- Pourquoi est-ce que g ne peut pas être du type :'a $\rightarrow$ bool ?
- Pour voir la réponse il faut expliciter les quantificateur des variables de types.

Quantification de variables de types

- Toutes les variables dans un type sont quantifiées universellement au début du type :
- Par exemple : Un type comme

$$a \rightarrow b \rightarrow a \times (b \rightarrow a)$$

est à lire comme

$$\forall a, b : \quad a \rightarrow b \rightarrow a \times (b \rightarrow a)$$

- Les variables a, b peuvent être instanciées par des types quelconques.

Quantification de variables de types

- Reprenons l'exemple :

`let f g = (g 42) && (g "coocoo")`

- Le type qu'on essaye de construire ici est :

$$(\forall a : \quad a \rightarrow \texttt{bool}) \rightarrow \texttt{bool}$$

- Des types avec des $\forall$ sous une flèche n'existent pas en OCaml, l'inférence a raison de refuser cette définition. Ce qui existe en OCaml est le type

$$\forall a : \quad ((a \rightarrow \texttt{bool}) \rightarrow \texttt{bool})$$

mais c'est un type différent !

Quantification de variables de types

- *Normalement*, dans tous les types les variables de types sont (implicitement) quantifiées $\forall$, avec un quantificateur devant le type complèt.
- *Normalement*, les variables de types libres (non quantifiées) paraissent seulement pendant la résolution des équations, une fois le type le plus général obtenu les variables sont quantifiées.
- Nous verrons une exception à cette règle un peu plus tard.
- Dans un type dérivé pour f dans **let** f x1 ... xn =e, toutes les nouvelles variables de type sont quantifiées au début par $\forall$.

Example : Quantification de variables de type (1)

- $\texttt{let } f\ g\ h = \underbrace{\texttt{fun } x \rightarrow \overbrace{h\ \underbrace{(g\ x)}_{t_3}}^{t_2}}_{t_1}$
- On obtient le système d'équations :

$$\begin{aligned} t_f &= t_g \rightarrow (t_h \rightarrow t_1) \\ t_1 &= t_x \rightarrow t_2 \\ t_h &= t_3 \rightarrow t_2 \\ t_g &= t_x \rightarrow t_3 \end{aligned}$$

Example : Quantification de variables de type (2)

- La solution trouvée pour la variable t_f est :

$$t_f = (t_x \rightarrow t_3) \rightarrow (t_3 \rightarrow t_2) \rightarrow (t_x \rightarrow t_2)$$

- Dans ce type, les variables t_x, t_2, t_3 sont libres, elles sont donc implicitement quantifiées avec un $\forall$
- Le type obtenu pour f est donc à lire comme :

$$\forall t_x, t_3, t_2 : (t_x \rightarrow t_3) \rightarrow (t_3 \rightarrow t_2) \rightarrow (t_x \rightarrow t_2)$$

Quelques résultats fondamentaux

- Il existe un algorithme qui, étant donnée une expression e, trouve, si elle est typable, son type σ *le plus général possible*, aussi appelé *type principal*.
- Le premier algorithme pour cela est le W de *Damas et Milner*, qu'on trouve dans *Principal type-schemes for functional programs. 9th Symposium on Principles of programming languages (POPL'82)*.
- Cet algorithme utilise de façon essentielle l'algorithme d'unification de Herbrand/Robinson.
- Les algorithmes modernes utilisent plutôt directement la résolution de contraintes.
- À la surprise générale, en 1990 on a montré que l'inférence de type pour le noyau de ML est DEXPTIME complète.

Exemples (inf4.ml)

```
let p x y = fun z -> z x y;;

let x0 = fun x -> x in
let x1 = p x0 x0 in
let x2 = p x1 x1 in
let x3 = p x2 x2 in
let x4 = p x3 x3 in
let x5 = p x4 x4 in
let x6 = p x5 x5 in
let x7 = p x6 x6 in
x7;;
```

Comment faire exploser le typeur d'OCaml...

- Dans l'exemple du transparent précédent : le type de `xn` a taille 2^n !
- ```
  let f0 = fun x -> (x, x) in
  let f1 = fun y -> f0 (f0 y) in
  let f2 = fun y -> f1 (f1 y) in
  let f3 = fun y -> f2 (f2 y) in
  let f4 = fun y -> f3 (f3 y) in
  f4 (fun z -> z);;
  ```
- Heureusement, en pratique, personne n'écrit de code comme ça.
- L'inférence de type à la ML reste un des système de type les plus puissants.

Les règles de typage pour OCaml : ...

- Standard :
  - sommes
  - tuples
  - enregistrements
- Plus compliqué : *value restriction*
  - typage des effets de bord (ce chapitre)
- Avancé :
  - modules
  - récursion polymorphe
  - objets
  - *variants polymorphes* (voir la semaine prochaine)
  - *GADT* (voir dans deux semaines)

La value restriction

- En OCaml, on dispose de structures mutables capables de contenir des données de tout type.
- Opérateurs pour les références :

ref	$\forall a : a \to (a\ ref)$	créer une référence vers une valeur
!	$\forall a : (a\ ref) \to a$	déréférencer
:=	$\forall a : (a\ ref) \to a \to unit$	changer la valeur d'une case mémoire référencée

- On peut imaginer que `ref` est un constructeur de type (au même titre que $\to$, $\times$, `list`, etc.)
- Essayons d'appliquer notre algorithme d'inférence de types en présence de références.

Exemples (inf6.ml)

```
let c = ref (function x -> x);;

c := (function x -> x+1);;

!c true;;
```
```

Inférence de types en présence de références

- Selon notre algorithme, `ref (function x -> x)` a le type

$$(a \to a)\ ref$$

- Donc, on obtient pour c le type :

$$\forall a : (a \to a)\ ref$$

- Le quantificateur universel pour a va permettre d'instancier a une fois par `int`, et puis par `bool`.
- Évidement OCaml a raison de refuser ce code, il y a donc un problème avec notre inférence de types.

Restriction de quantification

- Le malheur vient du fait que la variable 'a est quantifiée avec un $\forall$.
- En vérité, une fois la variable 'a instanciée, on ne devrait plus avoir le droit de changer cette instanciation.
- En présence de traits impératifs, on ne peut donc pas quantifier les variables dans les types comme avant.
- Idée : les variables de types sont quantifiées seulement quand l'expression à la droite du `let` satisfait certaines conditions, sinon la variable reste *libre* et peu donc être instanciée une seule fois.
- Ces conditions restent à déterminer !
- OCaml affiche une variable de type libre comme '_a.

Inférence de type en présence de référence

- Quelles sont les conditions qui permettent de quantifier les variables de type ?
- Une première idée est : l'expression ne contient pas du tout de références.
- Ca marche, mais a une conséquence assez grave : le polymorphisme est effectivement desactivé dès qu'on utilise des références dans une fonction, même si l'utilisation est parfaitement sûre.
- Regardons l'exemple suivant :

Exemples (value-restriction3.ml)

```
let fastrev = function list ->
  let left = ref list
  and right = ref []
  in begin
    while !left <> [] do
      right := (List.hd (!left)) :: !right;
      left :=  List.tl (!left)
    done;
    !right
  end;;

(* OK ! *)
fastrev [1;2;3;4];;
fastrev [true;true;false;false];;
```

Inférence de type en présence de référence

- La question est alors : trouver la bonne condition sous laquelle les variables de type peuvent être quantifiées, tel que :
  - les erreurs de type pendant l'excution du programme sont exclues ;
  - les fonctions polymorphes utilisant les références de façon sûre restent autorisées.
- Cette question a donné lieu à plusieurs propositions, toutes assez complexes.
- Une solution simple a été trouvée par Andrew K. Wright en 1995.

La *Value Restriction*

Solution simple introduite par SML : permettre la généralisation *seulement* pour les valeurs (d'où le nom). Les valeurs sont :

- les constantes (13, "foo", 13.0, ...)
- les variables (x, y, ...)
- les fonctions (fun x -> e), où *e* une expression *quelconque*
- les constructeurs appliqués à des valeurs (Foo v), excepté **ref**
- une valeur avec une contrainte de type (v : t)
- un n-uplet de valeurs (v1, v2, ...)
- un enregistrement contenant seulement des valeurs {l1 = v1, l2 = v2, ...}
- une liste de valeurs [v1, v2, ...]

Exemples (value-restriction1.ml)

```
let c = ref (function x -> x);;
  (* type of c: ('_a -> '_a) ref , '_a is a free variable! *)

c := (function x -> x+1);;
c;;
  (* type of c: (int -> int) ref, '_a has been instanciated *

!c true;;
  (* type error : clash between int and bool *)
```


Conséquences de la Value Restriction

- Dans l'exemple précédent, *c* a le type ('_a -> '_a) **ref**
- Explication : l'expression n'est pas une valeur, donc il n'y a pas de généralisation ($\forall$) lors du let.
- Inconvénient : il y a parfois des programmes correctes qui sont refusés, comme sur l'exemple suivant.
- On peut souvent contourner le problème.
- OCaml utilise une solution légèrement plus générale, due à Jacques Garrigue.

Exemples (value-restriction2.ml)

```
let id = function x -> x;;

(* type error *)
let f = id id;;
f 42;;
f "coocoo";;

(* solution: eta-expansion *)
let g = function x -> (id id) x;;
g 42;;
g "coocoo";;
```


Limites de la méthode I

```
let twice_only f =
  (* yields a variant of f that can be applied twice *)
  (* only, and that behaves like identity after that.*)
  let counter = ref 0
  in function x ->
               counter := !counter+1;
               if !counter <= 2 then f x else x
;;

(* the function double is not polymorphic   *)
(* since the ride-hand side is not a value. *)
let double = twice_only (function x -> x@x);;
;;

double [1; 2];;
double [1; 2];;
```


Limites de la méthode II

```
double [1; 2];;
double ["a"; "b"];;

(* Using eta-expansion we get a polymorphic   *)
(* function, but it does not behave the same! *)
(* At each application of double_eta, a new   *)
(* counter is created. *)

let double_eta =
  function y -> twice_only (function x -> x@x) y
;;

double_eta [1; 2];;
double_eta [1; 2];;
double_eta [1; 2];;
double_eta ["a"; "b"];;
```


Pour en savoir plus


Harry G. Mairson.
Deciding ML typability is complete for deterministic exponential time.
In Proceedings of the 17th ACM SIGPLAN-SIGACT symposium on Principles of programming languages, POPL '90, pages 382–401, New York, NY, USA, 1990. ACM.


Andrew K. Wright.
Simple imperative polymorphism.
Lisp Symb. Comput., 8(4) :343–355, December 1995.


Jacques Garrigue.
Relaxing the value restriction.
In FLOPS 2004, pages 196–213.